GAMS Basics

Stefan Vigerske stefan@gams.com



Some of the following material is based on the GAMS lecture material by Josef Kallrath.

Outline

Basic Modeling

Dynamic Models

More GAMS

Outline

Basic Modeling

Dynamic Models

More GAMS

Basic Modeling 3 / 56

Cows & Pigs Example

Variables:

 x_1 the number of cows to purchase

 x_2 the number of pigs to purchase

Objective:

maximize
$$z = 3x_1 + 2x_2$$



Constraints:

$$x_1 \in \{0, 1, 2\}$$
$$x_2 \in \{0, 1, 2\}$$
$$x_1 + x_2 \le 3.5$$

File: cowspigs.gms

Basic Modeling 4 / 5

Variables

Syntax: [var-type] variable[s] varname [text] {, varname [text]}

var-type allows to pre-determine the Range of a variable:

```
Variable type Range free (default) \mathbb{R} positive \mathbb{R}_{\geq 0} negative \mathbb{R}_{\leq 0} binary \{0,1\} integer \{0,1,\ldots,100\} (!!) semicont \{0\}\cup\{\ell,u] (default: \ell=1,\,u=\infty) semiint \{0\}\cup\{\ell,ell+1,\ldots,u\} (default: \ell=1,\,u=100) sos1, sos2 special ordered sets of type 1 and 2 (later ...)
```

► Examples:

```
variables x, y, objvar;
positive variable x;
integer variable z;
```

Basic Modeling 5 / 56

Variable Attributes

Attributes of a variable:

Attribute	Meaning
.lo	lower bound on variable range
.up	upper bound on variable range
.fx	fixed value for a variable
.1	current primal value (updated by solver)
.m	current dual value (updated by solver)
.scale	scaling factor
.prior	branching priority

► Examples:

```
x.up = 10;
y.fx = 5.5;
display z.l;
```

Basic Modeling 6 / 56

Equations

► Equations serve to define restrictions (constraints) and an objective function

Declaration:

```
► Syntax: Equation[s] eqnname [text] {, eqnname [text]};
```

Example:

e1

```
Equation objective "Objective Function"
                    "First constraint"
```

Basic Modeling 7 / 56

Equations

▶ Equations serve to define restrictions (constraints) and an objective function

Declaration:

- ► Syntax: Equation[s] eqnname [text] {, eqnname [text]} ;
- Example:

```
Equation objective "Objective Function" e1 "First constraint"
```

Definition:

Syntax: eqnname(domainlist).. expression eqn_type expression;

Example:

```
objective.. objvar =e= 2 * x + 3 * y*y - y + 5 * z ;
e1.. x + y =l= z;
```

Basic Modeling 7 / 56

Equation Attributes

Attributes of an equation:

Attribute	Meaning
.1	activity (updated by solver)
.m	current dual value (updated by solver)
.scale	scaling factor

Basic Modeling 8 / 5

Model statement

- ► Model = a collection of equations
- Syntax:

Example:

```
model m / all /;
model m / objective, e1 /;
```

Basic Modeling 9 / 56

Model statement

- ► Model = a collection of equations
- Syntax:

► Example:

```
model m / all /;
model m / objective, e1 /;
```

Attributes:

set by user		set by solver	
iterlim ite	eration limit	iterusd	number of iterations
optcr re	me limit in seconds lative gap tolerance	resusd modelstat	solving time model status
optfile nu	ımber of solver options file	solvestat	solver status

Example:

Basic Modeling

```
m.reslim = 60;     m.optcr = 0;
solve m minimize objvar using MINLP;
display m.resusd;
```

Solving a model

- ▶ Passing a model to a solver and evaluation of results
- Specification of one free variable to be minimized or maximized
- Syntax:

```
solve modelname using modeltyp maximizing minimizing varname; solve modelname maximizing minimizing varname using modeltyp;
```

▶ the model type defines the problem class to be used for the model:

```
LP
          a linear problem
QCP
          quadratically constraint problem (only linear and quadratic terms)
NI.P
          a nonlinear problem with continuous functions
DNLP
          a nonlinear problem with discontinuous functions
MTP
          a mixed-integer linear problem
          a mixed-integer quadratically constraint problem
MIQCP
          a mixed-integer nonlinear problem
MTNI.P
          a nonlinear constraint satisfaction problem (no objective function)
CNS
          a mixed-integer problem with relaxed integrality restrictions
RMIP,...
```

Example:

```
solve m using MINLP minimizing objvar; solve m using RMINLP min objvar;
```

Basic Modeling 10 / 56

GAMS command line call

Calling GAMS from the command line:

```
$ gams <modelfile> { [-]key=value | key value }
```

Examples:

- gams trnsport.gms
- ▶ gams trnsport
- ▶ gams trnsport.gms LP=CBC
- ▶ gams trnsport LP CBC
- ▶ gams trnsport -LP CBC

Basic Modeling 11 / 56

Listing File

Running a GAMS model generates a listing file (.1st file).

Compilation Errors:

- ▶ are indicated by ****
- contain a '\$' directly below the point at which the compiler thinks the error occurred
- ▶ are explained near the end of the line-numbered listing part
- ▶ in the IDE, they are also indicated by red lines in the process (log) window (can be double-clicked)
- ▶ check carefully for the cause of the first error, fix it, and try again
- usual causes: undefined / undeclared symbols (parameters, variables, equations), unmatched brackets, missing semi-colons

Basic Modeling 12 / 56

Listing File: Equation and Column Listing

Equation Listing:

- ▶ listing of generated equations with sets unrolled, parameters removed, ...
- useful for model debugging: is the intended model generated?
- ▶ for nonlinear equations, a linearization in the starting point is shown

```
AcidDef.. AcidDilut*AcidErr =e= 35.82-22.2*F4Perf;

-> AcidDef.. (1)*AcidDilut + 22.2*F4Perf + (3.6)*aciderr =E= 35.82 ; (LHS = 35.79, INFES = 0.03 ****)
```

activity and violation of constraint in starting point also shown

Basic Modeling 13 / 56

Listing File: Equation and Column Listing

Equation Listing:

- ▶ listing of generated equations with sets unrolled, parameters removed, ...
- ▶ useful for model debugging: is the intended model generated?
- ▶ for nonlinear equations, a linearization in the starting point is shown

```
AcidDef.. AcidDilut*AcidErr =e= 35.82-22.2*F4Perf;

-> AcidDef.. (1)*AcidDilut + 22.2*F4Perf + (3.6)*aciderr =E=

35.82 ; (LHS = 35.79, INFES = 0.03 ****)
```

► activity and violation of constraint in starting point also shown

Column Listing:

- ▶ shows coefficients, bounds, starting values for generated variables
 - -- F4Perf F4 Performance Number

```
F4Perf
```

```
(.LO, .L, .UP, .M = 1.45, 1.45, 1.62, 0) AcidDef
```

(1) F4Def

22.2

Basic Modeling 13 / 56

Listing File: Solve Summary

- generated for each solve command
- reporting status and result of solve

```
SOLVE
                             S U M M A R Y
                                OBJECTIVE
    MODEL
            m
    TYPE
                                DIRECTION
            NLP
                                           MINIMIZE
    SOLVER
            CONOPT
                                FROM LINE
                                           85
**** SOLVER STATUS
                      1 Normal Completion
**** MODEL STATUS
                      2 Locally Optimal
**** OBJECTIVE VALUE
                                  -1.7650
RESOURCE USAGE, LIMIT
                               0.006
                                          1000.000
ITERATION COUNT, LIMIT
                              16
                                    2000000000
EVALUATION ERRORS
                               0
```

Basic Modeling 14 / 56

Listing File: Solution Listing

- equation and variable primal and dual values and bounds

marking of infeasibility'.' = zero	ilities, "non-opti	malities", and	l unboundedness
	LOWER	LEVEL	UPPER
EQU Objective			

MARGINAL 1,0000 -- EQU AlkylShrnk -4.6116

-- EQU AcidBal -0.0020 -- EQU IsobutBal 0.0952 0.0127

-- EQU AlkylDef 0.5743 0.5747

-- EQU OctDef -- EQU AcidDef 35.8200 -1.3300

35.8533 -1.3300

-- EQU F4Def

LOWER.

-TNF -- VAR F -- VAR OlefinFeed

0.8500

-1.4143

1.6198 1.3617 0.7185

2.8790

1.8926

0.8998

LEVEL

+INF 2.0000 1.6000

0.5743

35.8200

UPPER

1.2000

5.0000

2,0000

0.9300

-1.3300

-0.1269NOPT -0.2133 NOPT -0.0411 NOPT

-4.1992

MARGINAL

11.8406 INFES

0.0563 INFES

-1.0763 INFES

-25.9326 INFES

0.2131 INFES

-0.0076 NOPT -0.4764 NOPT 0.5273 NOPT

Basic Modeling

-- VAR IsobutRec

-- VAR AcidFeed

-- VAR AlkylYld

-- VAR IsobutMak

-- VAR AcidStren

15 / 56

Cows & Pigs Continued

Define:

- a set of animals (i)
- a parameter with profit for each animal (p(i))
- a parameter with maximal number for each animal (xmax(i))
- a parameter for the max number of all animals (maxanimal)
- ▶ an integer variable to count how many of each animal to buy (x(i))
- a real variable to hold the profit (profit)
- an equation to define the objective
- ▶ an equation to limit the total number of purchased animals by maxanimal

Fill sets and parameters with the data from the original example:

- 2 animals: cow and pig
- profit for cow: 3 profit for pig: 2
- maximal number of cows: 2 maximal number of pigs: 2
- maximal number of animals: 3.5

File: cowspigs2.gms

Basic Modeling 16 / 56

Sets

- Basic elements of a model
- ► Syntax:

```
set set_name ["text"] [/element ["text"] {,element ["text"]} /]
    {,set_name ["text"] [/element ["text"] {,element ["text"]} /]};
```

Basic Modeling 17 / 56

Sets

- Basic elements of a model
- ► Syntax:

```
set set_name ["text"] [/element ["text"] {,element ["text"]} /]
   {,set_name ["text"] [/element ["text"] {,element ["text"]} /]};
```

- Name set_name is identifier
- ▶ Elements have up to 63 characters, start with letter or digit or are quoted:

```
A Phos-Acid September 1986 1952-53 Line-1 
'*TOTAL*' '10%incr' '12"/foot' "Line 1"
```

- ▶ Elements have no value (!), that is, '1986' does not have the numerical value 1986 and '01' \neq '1'
- ▶ Text has up to 254 characters, all in one line

Basic Modeling 17 / 56

Sets

- Basic elements of a model
- ► Syntax:

```
set set_name ["text"] [/element ["text"] {,element ["text"]} /]
   {,set_name ["text"] [/element ["text"] {,element ["text"]} /]};
```

- Name set_name is identifier
- ▶ Elements have up to 63 characters, start with letter or digit or are quoted:

```
A Phos-Acid September 1986 1952-53 Line-1 '*TOTAL*' '10%incr' '12"/foot' "Line 1"
```

- ▶ Elements have no value (!), that is, '1986' does not have the numerical value 1986 and '01' \neq '1'
- ▶ Text has up to 254 characters, all in one line
- ► Example:

```
Set n Nutrients
/ Prot "Protein (mg)"
  VitA "Vitamine A", VitC 'Vitamine C',
  Calc Calcium
/;
```

Basic Modeling 17 / 56

Sequences, Alias

Sequences in Sets: *-Notation

```
set t "time" / 2000*2008 /;
corresponds to
set t "time" / 2000,2001,2002,2003,2004,2005,2006,2007,2008 /;
```

- ▶ a1bc*a20bc is different from a01bc*a20bc
- ▶ the following are wrong: a1x1*a9x9, a1*b9

Basic Modeling 18 / 56

Sequences, Alias

Sequences in Sets: *-Notation

- set t "time" / 2000*2008 /;
 corresponds to
 set t "time" / 2000,2001,2002,2003,2004,2005,2006,2007,2008 /;
- ▶ a1bc*a20bc is different from a01bc*a20bc
- ▶ the following are wrong: a1x1*a9x9, a1*b9

Several names for one set: alias command

- Syntax: alias(set_name, set_name \{, set_name\})
- Example:

```
set ice / chocolate, strawberry, cherry, vanilla /;
alias(ice, icecreme, gelado, sorvete);
```

Basic Modeling 18 / 56

Data

- data in GAMS consists always of real numbers (no strings)
- uninitialized data has the default value 0
- 3 forms to declare data:
 - Scalar: a single (scalar) date
 - ► Parameter: list oriented data
 - ► Table: table oriented data (at least 2 dimensions)

Basic Modeling 19 / 50

Data

- data in GAMS consists always of real numbers (no strings)
- uninitialized data has the default value 0
- 3 forms to declare data:
 - ► Scalar: a single (scalar) date
 - Parameter: list oriented data
 - ► Table: table oriented data (at least 2 dimensions)

Scalar Data:

Syntax:

► Example:

```
Scalars rho Discountfactor / .15 / izf internal rate of return:
```

Basic Modeling 19 / 56

Data: Parameters

Parameter:

- can be indexed over a one or several sets
- Syntax:

Example:

Example:

```
set c 'countries' / jamaica, haiti, guyana, brazil /;
parameter demand(c,ice) "Demand of icecream per country (t)"
/ Jamaica.Chocolate 300, Jamaica.Strawberry 50, Jamaica.Cherry 5
    Haiti.(Chocolate, Vanilla, Strawberry) = 30,
    (Guyana, Brazil).Chocolate 100 /
```

Basic Modeling 20 / 56

Data: Tables

Tables:

Syntax:

```
table table_name [text] EOL
        element { element }
    element signed_num { signed_num } EOL
    {element signed_num { signed_num } EOL} ;
```

Example:

```
table demand(c,ice) "Demand of icecream per country (t)"
Chocolate Strawberry Cherry Vanilla

Jamaica 300 50 5

Haiti 30 30 30

(Guyana, Brazilien) 100
```

- ▶ no "free form": position of elements is of importance
- ▶ tables with more than 2 dimensions are also possible

Basic Modeling 21 / 56

Data: Assignments

► Scalar Assignment:

```
scalar x / 1.5 /;
x = 1.2;
x = x + 2;
```

► Indexed Assignment:

```
Set row / r1*r10 /
   col / c1*c10 /
   subrow(row) / r7*r10 /:
Parameter a(row,col), r(row), c(col);
a(row,col) = 13.2 + r(row)*c(col);
a('r7', 'c4') = -2.36;
a(subrow, c10) = 2.44 - 33*r(subrow);
a(row,row) = 7.7 - r(row);
alias(row,rowp);
a(row, rowp) = 7.7 - r(row) + r(rowp);
```

Basic Modeling 22 / 56

Data: Expressions

Expression: an arbitrarily complex calculation instruction

```
Arithmetic operators: ** (exponentiate), +, -, *, /
```

- x = 5 + 4*3**2;x = 5 + 4*(3**2);
- ▶ x**n corresponds to $exp(n*log(x)) \Rightarrow only$ allowed for x > 0
- population(t) = 56*(1.015**(ord(t)-1))

Basic Modeling 23 / 56

Data: Expressions

```
Expression: an arbitrarily complex calculation instruction
Arithmetic operators: ** (exponentiate), +, -, *, /
 x = 5 + 4*3**2:
    x = 5 + 4*(3**2):
 \triangleright x**n corresponds to exp(n*log(x)) \Rightarrow only allowed for x > 0

ightharpoonup population(t) = 56*(1.015**(ord(t)-1))
Indexed Operations:
 Syntax: indexed_op( (controlling_indices), expressions)
 ▶ indexed_op can be: sum, prod, smin, smax
    parameter demand(c,ice) "demand (t)"
               totaldemand(c) "totaler demand per country (t)"
               completedemand "totaler demand for all countries (t)"
               mindemand(ice) "minimal demand per icecream";
    totaldemand(c) = sum(ice, demand(c,ice));
    completedemand = sum((ice,c), demand(c,ice));
    mindemand(ice) = smin(c, demand(c,ice));
```

Basic Modeling 23 / 56

Data: Expressions

```
Expression: an arbitrarily complex calculation instruction
Arithmetic operators: ** (exponentiate), +, -, *, /
```

- x = 5 + 4*3**2: x = 5 + 4*(3**2):
 - \triangleright x**n corresponds to exp(n*log(x)) \Rightarrow only allowed for x > 0
 - population(t) = 56*(1.015**(ord(t)-1))
 (+ -1)

Indexed Operations:

- Syntax: indexed_op((controlling_indices), expressions)
- ▶ indexed_op can be: sum, prod, smin, smax

```
parameter demand(c,ice) "demand (t)"
          totaldemand(c) "totaler demand per country (t)"
          completedemand "totaler demand for all countries (t)"
```

mindemand(ice) "minimal demand per icecream";

totaldemand(c) = sum(ice, demand(c,ice)); completedemand = sum((ice,c), demand(c,ice));

mindemand(ice) = smin(c, demand(c,ice));

Functions: errorf(x), exp(x), power(x,n), sqr(x), uniform(a,b), normal(mean, sdev), ...

What the Solve Statement is doing

At each Solve statement, the following happens...

- The model (= list of (indexed) equations) is compiled into a model instance, that is, a scalar (= no indices) list of constraints and those variables, that appear in at least one constraint. (The Equation Listing and Column Listing shows in the listing file shows parts of this model.)
- 2. Instance statistics are written to the log:

```
--- Generating NLP model m
```

- --- alkyl.gms(85) 5 Mb
- --- 8 rows 15 columns 32 non-zeroes
- --- 54 nl-code 19 nl-non-zeroes
- 3. Some error check is performed.
- 4. The instance is passed on to a solver and processed there.

```
--- Executing IPOPT: elapsed 0:00:00.093 [solver log]
```

- --- Restarting execution
- 5. The result (model and solution status, solution, statistical information) is passed back to GAMS and reported in the listing file.

Basic Modeling 24 / 56

Solver Status

Solver status: state of the solver program

```
Normal Completion
Iteration Interrupt
Resource Interrupt (timelimit reached)
Terminated by Solver
Evaluation Error Limit
Capability Problems
Licensing Problems
User Interrupt
setup, solver, system errors
```

Basic Modeling 25 / 56

Model Status

Model status: what the solution looks like

```
Optimal (global optimality)
2, 8
            Locally Optimal
      Unbounded
3, 18
4, 10, 19 Infeasible (proven to be infeasible)
5
            Locally Infeasible
            Intermediate Infeasible
6
            Intermediate Nonoptimal (feasible for continuous models)
9
            Intermediate Non-integer
11
            License Problem
12, 13
            Error
14
   No Solution (but expected so)
15, 16, 17 Solved for CNS models
```

Basic Modeling 26 / 56

Option command

- ▶ specification of systemwide parameters to control output, solving process, ...
- Syntax:

```
option keyword1 [ = value1 ] { ,|EOL keyword2 = [ value2 ] }
```

some important parameters:

keyword	meaning	default
iterlim	iteration limit	2000000000
reslim	time limit	1000 (!!)
optca	absolute gap tolerance	0.0
optcr	relative gap tolerance	0.1 (!!)
LP	choice of LP solver	CPLEX
NLP	choice of NLP solver	CONOPT

Example:

```
option iterlim = 100, optcr = 0;
solve icesale using mip min kosten;
option mip = cbc;
solve icesale using mip min kosten;
```

options can also be set on the command line:

> gams icesale.gms mip=cbc optcr=0

Basic Modeling 27 / 56

Display command

- display sets, data, variable/equation/model attributes in the listing file
- ► Examples:

```
display ice, sorbet;
display x.l, x.m;
display demand;
display satdemand.m;
```

- only non-zero values are displayed
- control number of digits after the decimal point for all displayed values option decimals = 1;
- number of digits after the decimal point for variable x: option x:6;

Basic Modeling 28 / 56

Outline

Basic Modeling

Dynamic Models

More GAMS

Dynamic Models 29 / 5

Cows & Pigs by Complete Enumeration (!) in GAMS

```
x_1 the number of cows to purchase (x_1 \in \{0,1,2\}) x_2 the number of pigs to purchase (x_2 \in \{0,1,2\}) maximize z=3x_1+2x_2 such that x_1+x_2 \leq 3.5
```

Dynamic Models 30 / 56

Cows & Pigs by Complete Enumeration (!) in GAMS

```
x_1 the number of cows to purchase (x_1 \in \{0, 1, 2\})
          x_2 the number of pigs to purchase (x_2 \in \{0, 1, 2\})
    maximize z = 3x_1 + 2x_2
    such that x_1 + x_2 < 3.5
scalar x1, x2, obj;
scalar objbest, x1best, x2best;
objbest = 0;
for (x1 = 0 to 2,
  for (x2 = 0 \text{ to } 2.
    if(x1 + x2 le 3,
       obj = 3*x1 + 2*x2;
       if (obj > objbest,
         x1best = x1:
         x2best = x2;
         objbest = obj;
   )))))
display x1best, x2best, objbest;
```

File: cowspigsenum.gms

Dynamic Models

Finding a good local optimum to a NLP: Multistart

- ► Starting an NLP solver from different starting points and pick the best solution.
- ▶ If we don't know how to pick a good point, let's pick one randomly.

Dynamic Models 31 / 56

Finding a good local optimum to a NLP: Multistart

- ► Starting an NLP solver from different starting points and pick the best solution.
- ▶ If we don't know how to pick a good point, let's pick one randomly.
- ► Multistart algorithm:
 - 1. $f^U = \infty$
 - 2. for k = 1 to N, do
 - 2.1 Generate starting point x uniformly at random over $[\underline{x}, \overline{x}]$.
 - 2.2 Run NLP solver from x and obtain solution x^* .
 - 2.3 if $f(x^*) < f^U$: $f^U = f(x^*)$ and $x^U = x^*$
 - 3. Return x^U and f^U .
- ▶ The GAMS solver MSNLP implements such an algorithm.


Dynamic Models 31 / 56

Finding a good local optimum to a NLP: Multistart

- Starting an NLP solver from different starting points and pick the best solution.
- ▶ If we don't know how to pick a good point, let's pick one randomly.
- Multistart algorithm:
 - 1. $f^U = \infty$
 - 2. for k = 1 to N, do
 - 2.1 Generate starting point x uniformly at random over $[\underline{x}, \overline{x}]$.
 - 2.2 Run NLP solver from x and obtain solution x^* . 2.3 if $f(x^*) < f^U$: $f^U = f(x^*)$ and $x^U = x^*$
 - 3. Return x^U and f^U .
- ► The GAMS solver MSNLP implements such an algorithm.
- Example:

$$\max_{x \in [0,10]} \phi(x),$$
 where $\phi(x) = (10 - x) \sin^9(2\pi x)$

- Optimal solution:
- x = 0.24971128, $\phi(x) = 9.75014433$ File: nlpsin.gms



Program flow control, loop command

- controlling the execution of a GAMS program
- ▶ commands: loop, if-else, while, for
- declarations and definition of equations are not allowed inside these commands
- solve statements are allowed

Dynamic Models 32 / 56

Program flow control, loop command

- controlling the execution of a GAMS program
- ▶ commands: loop, if-else, while, for
- declarations and definition of equations are not allowed inside these commands
- solve statements are allowed

loop command:

Example:

Dynamic Models 32 / 56

If-Elseif-Else command

```
Syntax:
  if( condition, statement {; statement};
  {elseif condition, statement {; statement}; }
  [else statement {; statement};]
  );
Example:
  if ((ml.modelstat eq 4),
         model ml was infeasible,
         relax bound and solve again
    x.up(j) = 2*x.up(j);
    solve ml using lp min z;
  elseif ml.modelstat eq 1,
    display x.l;
  else
    abort "Error solving the model";
  );
```

Dynamic Models 33 / 56

While, Repeat commands

While command:

```
Syntax: while(condition, statement \{; statement\}; );
Example:
    scalar count / 1 /;
    scalar globmin / inf /;
    while(count le 1000,
        x.l(j) = uniform(x.lo(j), x.up(j));
        solve ml using nlp min obj;
        if ( obj.l le globmin, globmin = obj.l; );
        count = count + 1;
    );
```

Dynamic Models 34 / 56

While, Repeat commands

While command:

- ► Syntax: while(condition, statement \{; statement\};);
- Example:
- scalar count / 1 /;
- scalar globmin / inf /;
 while(count le 1000,
 - x.l(j) = uniform(x.lo(j), x.up(j));
 - solve ml using nlp min obj;
 - if (obj.l le globmin, globmin = obj.l;);
 count = count + 1;
 -);

Repeat command:

- ► Syntax: repeat(statement \{; statement\}; until condition);
- Example:

```
scalar count / 1 /; scalar globmin / inf /;
```

count = count + 1;
until count eq 1000);

For command

```
for(i = start to|downto end [by incr], statement \{; statement\};);
 ▶ Note: i is a scalar, not a set
 start, end, and incr can be real numbers, but incr needs to be positive
 Example:
    scalar count;
    scalar globmin / inf /;
    for ( count = 1 to 1000.
      x.l(j) = uniform(x.lo(j), x.up(j));
      solve ml using nlp min obj;
      if (obj.l le globmin, globmin = obj.l; );
   );
```

Dynamic Models 35 / 56

Subsets, Cardinality

Subsets:

- Syntax for set_name1 ⊆ set_name2: set set_name1 (set_name2);
- Example:

```
set ice / chocolate, strawberry, cherry, vanilla /;
set sorbet(ice) / strawberry, cherry /;
```

▶ Domain checking: set sorbet(ice) / strawberry, banana /; ⇒ error

Dynamic Models 36 / 56

Subsets, Cardinality

Subsets:



- Syntax for set_name1 ⊆ set_name2: set set_name1 (set_name2);
- ► Example:

```
set ice / chocolate, strawberry, cherry, vanilla /;
set sorbet(ice) / strawberry, cherry /;
```

lacktriangle Domain checking: set sorbet(ice) / strawberry, banana /; \Rightarrow error

Card(set)

- gives the number of elements in a set
- Example:

```
set c 'countries' / jamaica, haiti, guyana, brazil /;
scalar nc 'number of countries;
nc = card(c);
```

Dynamic Models 36 / 56

Ordered Sets

Lag & Lead Operations:

- allow to access neighbors (next or further distant) of a elements in a priori explicitly specified ordered set
- ► Syntax: setelement ± n
- ▶ Note: x(setelement+n) is zero if position of setelement > card(set)-n
- Example:

```
Set t / 1*24 /;
Variables level(t), inflow(t), outflow(t);
Equation balance(t) couple fill levels of reservoir over time;
* implicitly assume an initial fill level of zero
balance(t).. level(t) =e= level(t-1) + inflow(t) - outflow(t);
```

Dynamic Models 37 / 56

Ordered Sets

Lag & Lead Operations:

- allow to access neighbors (next or further distant) of a elements in a priori explicitly specified ordered set
- ightharpoonup Syntax: setelement \pm n
- ▶ Note: x(setelement+n) is zero if position of setelement > card(set)-n
- Example:

```
Set t / 1*24 /;
Variables level(t), inflow(t), outflow(t);
Equation balance(t) couple fill levels of reservoir over time;
* implicitly assume an initial fill level of zero
balance(t).. level(t) == level(t-1) + inflow(t) - outflow(t);
```

Ord(setelement):

- gives position of an element in an a priori explicitly specified ordered sets
- Example:

```
Set t / 1*24 /;
Parameter hour(t);
hour(t) = ord(t);
```

Dynamic Sets

- dynamic sets allow elements to be added or removed
- dynamic sets are usually domain-checked, i.e., subsets
- ► Syntax: setname(othersetelement) = yes | no (add/remove single element)
- ► Syntax: setname(subset) = yes | no (add/remove another subset)
- Example:

set sorbet(ice);

```
Set highdemand(ice) ice creams with high demand;
highdemand(ice) = (demand(ice) >= 500);
```

most often used as controlling index in an assignment or equation definition Scalar sumhighdemand; sumhighdemand = sum(highdemand, demand(highdemand));

Dynamic Models 38 / 56

Example: refer to first/last period of discrete-time models

Set t / 1*24 /;

Sets tb(t) base period

tb(t) = (ord(t) = 1);

tn(t) non-base periods
tt(t) terminal period;

```
tn(t) = (ord(t) > 1);
tt(t) = (ord(t) = card(t));
Variables level(t), inflow(t), outflow(t);
Equations balance(t) couple fill levels of reservoir over time
           basebalance(t) define fill level for base period;
* only for time periods > base period
balance(tn(t)).. level(t) =e= level(t-1) + inflow(t) - outflow(t);
* only for base period
basebalance(tb).. level(tb) =e= 100 + inflow(tb) - outflow(tb);
* lower bound on fill level in terminal period
level.lo(tt) = 100;
Alternatively (but less readable):
equation basebalance; basebalance...
sum(tb, level(tb))
pynamia Medda.
ee= 100 + sum(tb, inflow(tb) - outflow(tb));
```

Outline

Basic Modeling

Dynamic Models

More GAMS

More GAMS 40 / 56

Multidimensional Sets

Multidimensional Sets:

- describing assignments (relations) between sets
- Example:

```
sets c 'countries' / jamaica, haiti, guyana, brazil /
    h 'harbors' / kingston, s-domingo, georgetown, belem /
set hc(h, c) harbor to country relation
    / kingston.jamaica, s-domingo.haiti
    georgetown.guyana, belem.brazil /;
```

More GAMS 41 / 56

sameas and diag

```
sameas(setelement, otherelement) and sameas(setelement, "text")
diag(setelement, otherelement) and diag(setelement, "text")
```

- sameas returns true if identifiers for given set elements are the same, or if identifier of one set element equals a given text
- sameas can also be used as a set
- ▶ diag is like sameas, but return 1 if true, and 0 otherwise
- Example:

```
sets ice1 / chocolate, strawberry, cherry, vanilla /
    ice2 / strawberry, cherry, banana /;
scalar ncommon;
ncommon = sum((ice1, ice2), diag(ice1,ice2));
ncommon = sum(sameas(ice1, ice2), 1);
```

More GAMS 42 / 56

Conditional expressions: \$ Operator

Boolean Operators:

- numerical operators: lt, <, le, <=, eq, =, ne, <>, ge, >=, gt, >
- logical operators: not, and, or, xor
- set membership: a(i) evaluates to true if and only if i is contained in the (sub)set a, otherwise false
- true corresponds to 1, false to 0

More GAMS 43 / 56

Conditional expressions: \$ Operator

Boolean Operators:

- numerical operators: lt, <, le, <=, eq, =, ne, <>, ge, >=, gt, >
- logical operators: not, and, or, xor
- set membership: a(i) evaluates to true if and only if i is contained in the (sub)set a, otherwise false
- ▶ true corresponds to 1, false to 0

\$-Operator:

- allows to apply necessary conditions
- \$(condition) can be read as "if condition is true"
- and

- Example:
 - "If b>1.5, then let a = 2." $\Rightarrow a\$(b > 1.5) = 2$;
 "If b>1.5, then let a = 2." $\Rightarrow a\$(b > 1.5) = 2$;
 - "If b>1.5, then let a= 2, otherwise let a= 0." \Rightarrow a = 2\$(b > 1.5);
- ▶ \$ on left side: no assignment, if condition not satisfied
- \$ on right side: always assignment, but term with \$ evaluates to 0, if condition not satisfied
- cannot be used in declarations

More GAMS 43 /

Applications for \$ Operator

Filtering in indexed operations:

```
parameter sorbetbalance;
set ice; set sorbet(ice);
sorbetbalance = sum(ice$sorbet(ice), price(ice)*purchase.l(ice));
sorbetbalance = sum(sorbet(ice), price(ice) * purchase.l(ice));
```

More GAMS 44 / 56

Applications for \$ Operator

Filtering in indexed operations:

```
parameter sorbetbalance;
set ice; set sorbet(ice);
sorbetbalance = sum(ice$sorbet(ice), price(ice)*purchase.l(ice));
sorbetbalance = sum(sorbet(ice), price(ice) * purchase.l(ice));
```

Conditioned indexed operations:

```
rho = sum(i\$(sig(i) ne 0), 1/sig(i) - 1);
```

More GAMS 44 / 56

Applications for \$ Operator

Filtering in indexed operations:

```
parameter sorbetbalance;
set ice; set sorbet(ice);
sorbetbalance = sum(ice$sorbet(ice), price(ice)*purchase.l(ice));
sorbetbalance = sum(sorbet(ice), price(ice) * purchase.l(ice));
```

Conditioned indexed operations:

```
rho = sum(i$(sig(i) ne 0), 1/sig(i) - 1);
```

Conditioned equations:

```
Equation satdemand(ice);
satdemand(ice)$sorbet(ice).. purchase(ice) =g= demand(ice);
```

Existence of variables in constraints:

```
variables x, y; parameter A; equation e;
e.. x + y$(A>2) =e= A;
```

No Variables in \$ condition

► The following DOES NOT WORK:

```
binary variable x; variable y; equation e1, e2; e1(x = 1).. y =1= 100; e2 ... y(x = 1) =1= 100;
```

- ▶ The value of x is decided by the solver, not by the GAMS.
- ► However, GAMS has to evaluate \$-operators when assembling an instance in the Solve statement.
- ► Instead, you have to reformulate:

```
e.. y =1= 100 * x + y.up * (1-x);
That is: x=1 \Rightarrow y \le 100; x=0 \Rightarrow y \le y.up.
```

More GAMS 45 / 56

Compilation vs. Execution

GAMS processed models in 2 phases:

Compilation Phase:

- ▶ reads the complete GAMS model and translate into GAMS specific byte code
- processes declarations (variables, equations, sets, parameters)
- execute all compile-time commands (next slides)

A Transportation Problem (TRNSPORT, SEQ=1)

► Listing file:

```
GAMS 24.1.2 r40979 Released Jun 16, 2013 LEX-LEG x86_64/Linux
```

- 3 16:03:38 Page 1
- Compilation

```
<echo of all processed lines>
```

```
COMPILATION TIME = 0.002 SECONDS 3 MB 24.1.2 r4
```

Execution Phase:

More GAMS

▶ executes commands in program: assignments, solve, loop, while, for, if, ...

Compile Time Commands

- ▶ introduced by \$ sign in the first column (!)
- ► Syntax: \$commandname argumentlist {commandname argumentlist}
- modify behavior of GAMS compiler, e.g., how to process the input, customizing output
- ▶ allow program flow control on compilation level: call external programs, include other files, if-else, goto, ...
- modifying and reading compile-time variables

Examples:

- define a title for your GAMS program \$Title A Transportation Problem
- define a section that contains only comments

\$onText

This problem finds a least cost shipping schedule that meets requirements at markets and supplies at factories.

\$offText

disable echoing of input lines in listing file \$offListing

Compile Time Variables

- compile-time variables hold strings
- ▶ their value is accessed to via the %variablename% notation
- ▶ values are assigned via \$set or \$eval commands or on the GAMS command line via double-dash-options: gams --variablename variablevalue
- ▶ for \$eval, the variable value string is interpreted as numerical expression
- Example:

```
$set N 10
$eval Nsqr %N% * %N%
set i / 1 * %N% /:
```

variants \$setLocal, \$setGlobal, \$evalLocal, \$evalGlobal allow to control the (file)scope of a variable

More GAMS 48 / 56

Compile Time Program Control: \$If

- ▶ \$If allows to do conditional processing
- Syntax:

```
$If [not] (exist filename | string1 == string2) new_input_line
```

only one-line clauses allowed (new_input_line can be on next line, though)

Examples:

```
$if exist myfile.dat $log "myfile.dat exists, yeah!"
scalar a;
$if    %difficulty% == easy a = 5;
$if not %difficulty% == easy a = 10;
```

More GAMS 49 / 56

Compile Time Program Control: \$If

- ▶ \$If allows to do conditional processing
- Syntax:

```
$If [not] (exist filename | string1 == string2) new_input_line
```

- only one-line clauses allowed (new_input_line can be on next line, though)
- Examples:

```
$if exist myfile.dat $log "myfile.dat exists, yeah!"
scalar a;
$if    %difficulty% == easy a = 5;
$if not %difficulty% == easy a = 10;
```

► \$IfThen-\$ElseIf-\$Else-\$Endif allows to control activity for a set of statements; Example:

```
scalar a;
$ifthen %difficulty% == easy
a = 5;
$else
a = 10;
$endif
```

More GAMS 49 / 56

Compile Time Program Control: \$Goto

- ▶ \$Goto-\$Label allows to skip over or repeat sections of the input
- Example:

```
scalar a / 5 /;
$if %difficulty% == easy $goto easy
a = 10;
$label easy
```

More GAMS 50 / 56

Executing Shell Commands

- \$Call passes a following string to the current shell and waits for the command to be completed
- if the string starts with a '=', the operating system is called directly, i.e., no shell is invoked
- ► Example:

```
$call "gamslib trnsport"
$call "=gams trnsport"
```

```
$if exist myfile.dat $call cp myfile.dat mycopy.dat
```

the errorLevel functions allows to check whether a previous command (e.g., \$call) executed without error:

```
$call "gamslib trnsport"
$call "gams trnsport"
$if errorlevel 1 $abort "ouch! - solving trnsport failed"
```

More GAMS 51 / 56

Writing text files

- ▶ \$Echo and \$onEcho-\$offEcho allows to write to text files
- Example:

```
$Echo "hello, world!" > myfile.txt
$OnEcho >> myfile.txt
ahoy-hoy!
$OffEcho
```

- > myfile.txt creates a new file myfile.txt, thereby overwriting a possibly existing one of the same name
- >> myfile.txt appends to an existing file myfile.txt
- ▶ Recall: These are compilation-time commands! Not usable to write solve outcomes or similar; use display command or put-facility (later) for this.

More GAMS 52 / 56

Including text files

- ▶ \$Include allows to include ASCII files into a GAMS program
- compilation is then continued for the included file
- ► Example:

```
Parameter d(i,j) distance in thousands of miles; $include dist.inc
```

where dist.inc contains

Table d(i,j) distance in thousands of miles

	new-york	chicago	topeka
seattle	2.5	1.7	1.8
san-diego	2.5	1.8	1.4 ;

More GAMS 53 / 56

Including csv files

▶ \$OnDelim enables comma separated value (csv) format for data in table or parameter statements

Examples:

```
Table d(i,j) distance
                                       Parameter d(i,j) distance /
  $ondelim
                                       $ondelim
  $include dist.csv
                                       $include dist.txt
  $offdelim
                                       $offdelim
                                       /;
where dist.csv is
                                     where dist.txt is
  ,new-york,chicago,topeka
                                       SEATTLE, NEW-YORK, 2.5
  seattle, 2.5, 1.7, 1.8
                                       SAN-DIEGO, NEW-YORK, 2.5
  san-diego, 2.5, 1.8, 1.4
                                       SEATTLE, CHICAGO, 1.7
                                       SAN-DIEGO, CHICAGO, 1.8
```

More GAMS 54 / 56

SEATTLE, TOPEKA, 1.8 SAN-DIEGO, TOPEKA, 1.4

Put Command

- writing text files at execution time
- ▶ associating an identifier fileid with a file: file fileid / myfile.txt /;
- select a stream (and thus a file) to write to: put fileid;
- writing some items (text, labels, numbers): put item {, item};
- write a linebreak: put /;
- close a stream: putclose;

More GAMS 55 / 56

Put Command

```
writing text files at execution time
associating an identifier fileid with a file: file fileid / myfile.txt /;
select a stream (and thus a file) to write to: put fileid;
writing some items (text, labels, numbers): put item {, item};
write a linebreak: put /;
close a stream: putclose;
Example:
   file fx /result.txt/;
   put fx 'Shipped quantities between plants and markets' /;
          ·----·, /;
   put
   loop((i,j)$x.l(i,j),
     put 'Shipment from ', i.te(i):10, ' to ', j.te(j):10,
         ' in cases:', x.l(i,j) /;
   ); putclose;
  gives
   Shipped quantities between plants and markets
   Shipment from seattle to new-york
                                                 50.00
                                       in cases:
   Shipment from seattle to chicago in cases: 300.00
   Shipment from san-diego to new-york in cases: 275.00
   Shipment from san-diego to topeka
                                       in cases:
                                                 275.00
```

Label names and explanatory texts can be accessed via attributes:

- ▶ ident.ts: text associated with identifier
- ▶ element.tl: label associated with set element
- ▶ set.te(element): text associated with element of set

More GAMS 56 / 50

Label names and explanatory texts can be accessed via attributes:

- ▶ ident.ts: text associated with identifier
- ▶ element.tl: label associated with set element
- ▶ set.te(element): text associated with element of set

Local Item Formatting:

- Syntax for formatting item output: item:{<>}width:decimals
- ▶ {<>} specifies whether justified left (<), right (>), or centered (<>)
- width is the field width
- decimals is the number of decimals for numeric output
- each can be omitted, e.g., x.1::5

More GAMS 56 / 56

Label names and explanatory texts can be accessed via attributes:

- ▶ ident.ts: text associated with identifier
- ▶ element.tl: label associated with set element
- ▶ set.te(element): text associated with element of set

Local Item Formatting:

- ▶ Syntax for formatting item output: item:{<>}width:decimals
- ▶ {<>} specifies whether justified left (<), right (>), or centered (<>)
- width is the field width
- decimals is the number of decimals for numeric output
- ▶ each can be omitted, e.g., x.1::5

Global Item Formatting:

- change field justification and width for all items of a type
- ▶ .lj, .nj, .sj, .tj, .lw, .nw, .sw, .tw attributes of stream identifier

see GAMS User's Guide Section 15.10

More GAMS 56 / 56

Label names and explanatory texts can be accessed via attributes:

- ▶ ident.ts: text associated with identifier
- ▶ element.tl: label associated with set element
- ▶ set.te(element): text associated with element of set

Local Item Formatting:

- Syntax for formatting item output: item:{<>}width:decimals
- {<>} specifies whether justified left (<), right (>), or centered (<>)
- ▶ width is the field width
- decimals is the number of decimals for numeric output
- ▶ each can be omitted, e.g., x.1::5

Global Item Formatting:

- change field justification and width for all items of a type
- ▶ .lj, .nj, .sj, .tj, .lw, .nw, .sw, .tw attributes of stream identifier
- see GAMS User's Guide Section 15.10

Cursor Positioning:

▶ put @n; moves cursor to column n of current line

More GAMS